Algebraic and Arithmetic Attributes of Hypergeometric Functions in SageMath

Xavier Caruso and Florian Fürnsinn

Abstract

We report on implementations for algorithms treating algebraic and arithmetic properties of hypergeometric functions in the computer algebra system SageMath. We treat hypergeometric series over the rational numbers, over finite fields, and over the p-adics. Among other things, we provide implementations deciding algebraicity, computing valuations, and computing minimal polynomials in positive characteristic.

The second named author was funded by a DOC Fellowship (27150) of the Austrian Academy of Sciences at the University of Vienna. Further he thanks the French–Austrian project EAGLES (ANR-22-CE91-0007 & FWF grant 10.55776/I6130) for financial support. The authors thank Austria's Agency for Education and Internationalisation (OeAD) and Campus France for providing funding for research stays via WTZ collaboration project/Amadeus project FR02/2024.

1 Introduction

A *hypergeometric function* with rational *top parameters* $\boldsymbol{\alpha} \coloneqq (\alpha_1, \dots, \alpha_n) \in \mathbb{Q}^n$ and *bottom parameters* $\boldsymbol{\beta} \coloneqq (\beta_1, \dots, \beta_m) \in \mathbb{Q}^m$ is defined as the power series

$$ {}_nF_m\left(\boldsymbol{\alpha}, \boldsymbol{\beta}; x\right) \coloneqq \sum_{k=0}^{\infty} \frac{(\alpha_1)_k \cdots (\alpha_n)_k}{(\beta_1)_k \cdots (\beta_m)_k} \cdot \frac{x^k}{k!} \in \mathbb{Q}[\![x]\!], \tag{1} $$

where $(\gamma)_k \coloneqq \gamma(\gamma+1)\cdots(\gamma+k-1)$ denotes the *rising factorial* or *Pochhammer symbol*.

Hypergeometric functions exhibit very interesting properties. First of all, they are always *D-finite*, which means that they are solution of a linear differential equation with coefficients in $\mathbb{Q}[x]$. Besides, although they are themselves rarely algebraic over $\mathbb{Q}(x)$ (the sets of parameters for which this occurs have been fully classified) but their reductions modulo p are algebraic over $\mathbb{F}_p(x)$ whenever they are defined. These properties make them ideal test cases for conjectures about algebraic and arithmetic properties of D-finite series. The need to use a software for formulating and checking such conjectures appeared to the authors when they worked, jointly with Vargas-Montoya, on Galois groups of reductions modulo primes of D-finite series [6].

The present article is the outcome of this observation. It reports on an open source SageMath package designed to manipulate hypergeometric functions in various situations: over the rationals, over finite fields and over p-adic fields. To provide these functionalities it was necessary to develop new algorithms. They are presented mostly in [5].

Our package has been integrated in the standard release of SageMath since version 10.10.beta2 [4]. In this paper, we shortly showcase the main features of our package. Our presentation is complemented by an interactive worksheet available at:
https://xavier.caruso.ovh/notebook/hypergeometric-functions/
This interactive worksheet supports editing the examples and exploring new cases.

For the complete documentation of our package, we refer to the Section *Hypergeometric functions over arbitrary rings* in the reference manual of SageMath, available online here.

2 Setup

We propose an implementation of algebraic and arithmetic properties of hypergeometric functions, accessed by creating them by passing an argument in (the fraction field of) a polynomial ring or a power series ring to the standard constructor `hypergeometric`.

```
 In:  S.<x> = QQ[]
 In:  f = hypergeometric([1/9, 4/9, 5/9], [1/3, 1], x)
 In:  f
Out:  hypergeometric((1/9, 4/9, 5/9), (1/3, 1), x)
 In:  f.parent()
Out:  Hypergeometric functions in x over Rational Field
```

The main functions we will use throughout the following demonstration are $f(x)$ (defined above), which is Christol's example of a hypergeometric function that is not known to be a diagonal; $g(x)$, where $g(16x^2)$ is the generating function of Gessel excursions, and the somewhat obscure function $h(x)$ that illustrates many phenomena.

```
In:  g = hypergeometric([1/2, 5/6, 1], [5/3, 2], x)
In:  h = hypergeometric([1/5, 1/5, 1/5, 1/5], [1/3, 3^10/5 - 1], x)
```

3 Hypergeometric functions over $\mathbb{Q}$

Primes With Good Reductions We say that a series $s(x) \in \mathbb{Q}[\![x]\!]$ has good reduction at p, if all its coefficients have denominator coprime with p. For large enough prime numbers, this property only depends on the congruence class of p modulo a common denominator of the parameters. The method `good_reduction_primes()` computes this and outputs a set of prime numbers depending on congruence classes (implemented in SageMath by the first author [2]).

```
 In:  g.good_reduction_primes()
Out:  Set of all prime numbers with 2 excluded: 3, 5, 7, 11, ...
 In:  h.good_reduction_primes()
Out:  Set of prime numbers congruent to 1, 8, 11 modulo 15 with 17, 167, 677, 857, ..., 29327,
       29387 included and 23, 83, 113, 173, ..., 58913, 58943 excluded:  11, 17, 31, 41, ...
```

Details about the implemented algorithm, complexity statements and explicit bounds on the size of exceptional primes can be found in [5, §2, 3.1].

When almost all primes have good reduction, we say that the hypergeometric function is *globally bounded*; this can be directly checked using the method `is_globally_bounded()`.

Algebraicity All algebraic hypergeometric functions have been classified. We implemented the corresponding criterion [1, 8] in the method `is_algebraic()`.

```
 In:  f.is_globally_bounded()
Out:  True
 In:  h.is_globally_bounded()
Out:  False
```

```
 In:  f.is_algebraic()
Out:  False
 In:  g.is_algebraic()
Out:  True
```

4 Hypergeometric functions over $\mathbb{F}_p$

When a hypergeometric function $h(x)$ has good reduction at a prime p, we can form $h(x) \bmod p \in \mathbb{F}_p[\![x]\!]$ and study its properties. It is known for example that the latter is always algebraic [7, 10, 9, 6, 5].

Annihilating Polynomials An annihilating polynomial of a reduction of a hypergeometric function can be computed with the method `annihilating_ore_polynomial()`, which outputs a polynomial in the Frobenius morphism: to view it as an actual polynomial, one should replace $\texttt{Frob}^n$ by X^{p^n}.

```
 In:  f19.annihilating_ore_polynomial()
Out:  (18*x^76 + 13*x^57 + 6*x^38 + 17*x^19 + 12)*Frob^2 +
       (12*x^38 + 11*x^32 + ... + 18*x^12 + 7)*Frob + x^30 + 16*x^29 + ... + 6*x^13 + x^12
```

The algorithm we implemented is a small variation of the one described in [5, §3.3]. The polynomials obtained this way are never irreducible. By their nature as linearized polynomials, they are always divisible by X. However, in general, they will factor even further.

Congruences Modulo Primes It is possible that two hypergeometric functions with different set of parameters leads to series which are congruent modulo p, as showcased in the code example below.

```
 In:  T.<y> = GF(13)[]
 In:  h1 = hypergeometric([1/12, 1/4], [1/2], y)
 In:  h2 = hypergeometric([1/12, 1/6], [1/3], y)
 In:  h1.power_series(1000) - h2.power_series(1000)
Out:  O(y^1000)
```

The method `is_equal_as_series()` checks when this happens.

```
 In:  h1.is_equal_as_series(h2)
Out:  True
```

We sketch an algorithm to check when this holds in the longer version of this text [3].

5 Hypergeometric functions over $\mathbb{Q}_p$

Last, we deal with hypergeometric functions with rational parameters defined over the field of p-adic numbers $\mathbb{Q}_p$. For $x \in \mathbb{Q}_p$, we let $\mathrm{val}_p(x)$ denote its p-adic valuation and we let $\|x\|_p = p^{-\mathrm{val}_p(x)}$ be its p-adic norm.

Radius of Convergence Similar to the complex case, the p-adic radius of convergence of a series $s(x) = \sum a_k x^k \in \mathbb{Q}_p[\![x]\!]$ is defined as $\liminf_{k\to\infty} \|a_k\|_p^{-1/k}$. When $\|a\|_p$ (with $a \in \mathbb{Q}_p$) is less than this critical value, the series $s(a)$ converges in $\mathbb{Q}_p$. The method `log_radius_of_convergence()` computes the logarithm in base p of the p-adic radius of convergence of a hypergeometric series, cf. [5, §2].

Valuations For $\nu \in \mathbb{Q}$, we call $\mathrm{val}_{p,\nu}(s(x)) \coloneqq \min_{k\geq 0} \mathrm{val}_p(a_k) + \nu k$ the *ν-drifted p-adic valuation* of the series $s(x)$. `valuation()` computes the p-adic valuation of a hypergeometric function, and passing a parameter ν, it computes the ν-drifted p-adic valuation following [5, §2.2]. Additionally one can also pass the option `position=True`, to also output the minimal index k, for which the valuation is attained for the coefficient in x^k, cf. [5, Rem. 2.6].

p-adic Evaluations When a is a p-adic number with norm less than the radius of convergence, the value $h(a)$ makes sense. Our package allows to compute it using the obvious syntax, see [5, §2.4] for the algorithm used.

Newton Polygons The Newton polygon of a series $\sum_k a_k x^k$ is the convex hull in $\mathbb{R}^2$ of the points (k, v) with $v \geq \mathrm{val}_p(h_k)$.

For hypergeometric series, it can be computed with the method `newton_polygon()`. Passing an argument ν, with ν chosen smaller than the logarithmic p-adic radius of convergence, can help to handle cases where the Newton polygon has an infinite number of slopes.

```
 In:  hp5 = h.change_ring(Qp(5))
 In:  hp5.log_radius_of_convergence()
Out:  -7/2
 In:  fp5 = f.change_ring(Qp(5))
 In:  fp5.valuation()
Out:  0
 In:  hp3 = h.change_ring(Qp(3))
 In:  hp3.valuation(position=True)
Out:  (-4, 2)
```

```
 In:  hp5.valuation()
Out:  -Infinity
 In:  hp5.valuation(-7/2)
Out:  0
 In:  fp5(5)
Out:  1 + 3*5^2 + 5^4 + ... + O(5^20)
 In:  hp3(1/3)
Out:  3^-5 + 2*3^-1 + 1 + 2*3 + ... + O(3^13)
```

References

[1] F. Beukers and G. Heckman. 1989. Monodromy for the hypergeometric function ${}_nF_{n-1}$. *Inventiones Mathematicae*, 95, 2, 325–354. DOI: 10.1007/BF01393900.

[2] X. Caruso. 2025. Subsets of primes defined by congruence conditions. https://github.com/sagemath/sage/pull/41122.

[3] X. Caruso and F. Fürnsinn. 2026. Algebraic and Arithmetic Attributes of Hypergeometric Functions in SageMath. eprint: arXiv:2602.04531 (cs).

[4] X. Caruso and F. Fürnsinn. 2025. Algebraic and modular properties of hypergeometric functions. https://github.com/sagemath/sage/pull/41113.

[5] X. Caruso and F. Fürnsinn. 2026. Algorithms for algebraic and arithmetic attributes of hypergeometric functions. eprint: arXiv:2601.16105 (math.NT).

[6] X. Caruso, F. Fürnsinn, and D. Vargas-Montoya. 2025. Galois groups of reductions modulo p of D-finite series. eprint: arXiv:2504.09429 (math.NT).

[7] G. Christol. 1986. Fonctions et éléments algébriques. *Pacific Journal of Mathematics*, 125, 1, 1–37. DOI: 10.2140/pjm.1986.125.1.

[8] F. Fürnsinn and S. Yurkevich. 2024. Algebraicity of hypergeometric functions with arbitrary parameters. *Bulletin of the London Mathematical Society*, 2824–2846. DOI: 10.1112/blms.13103.

[9] D. Vargas-Montoya. 2024. Algebraicity modulo p of generalized hypergeometric series ${}_nF_{n-1}$. *Journal of Number Theory*, 259, 273–321. DOI: 10.1016/j.jnt.2024.01.004.

[10] D. Vargas-Montoya. 2021. Algébricité modulo p, séries hypergéométriques et structures de frobenius fortes. *Bulletin de la Société mathématique de France*. DOI: 10.24033/bsmf.2834.